4. přednáška (Datové typy v Maplu, chyby při výpočtech)

Datové typy

Pro zjišťování datového typu výrazu slouží příkaz **whattype**.

```
> whattype(230);
```

$$integer$$

```
> whattype(-234/129);
```

$$fraction$$

```
> whattype(Pi);
```

$$symbol$$

```
> whattype(sqrt(2));
```

$$\wedge$$

```
> whattype(1.2);
```

$$float$$

```
> whattype(x+y);
```

$$+$$

```
> whattype("Matematicka analyza s Maplem");
```

$$string$$

```
> M:=<<1,2>|<3,4>>;
```

$$M := \begin{bmatrix} 1 & 3 \\ 2 & 4 \end{bmatrix}$$

```
> whattype(M);
```

$$Matrix$$

```
> v:=Vector([1,2,3,4]);
```

$$v := \begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \end{bmatrix}$$

```
> whattype(v);
```

$$Vector_{column}$$

```
> v:=Vector[row]([1,2,3,4]);
```

$$v := [1, 2, 3, 4]$$

```
> whattype(v);
```

$$Vector_{row}$$

```
>
```

Datový typ posloupnost (exprseq)

```
> p:=5,4,3,2,1;
```

$$p := 5, 4, 3, 2, 1$$

```
> whattype(p);
```

$$exprseq$$

```
> p[2];
```

$$4$$

```
> p[-1];
```

$$1$$

```
>
```

Příkaz **seq** rovněž generuje datový typ posloupnost.

```
> p:=seq(i^2,i=1..10);
```

$$p := 1, 4, 9, 16, 25, 36, 49, 64, 81, 100$$

```
> whattype(p);
```

$$exprseq$$

```
> p[3];
```

$$9$$

```
> x$5;
```

$$x, x, x, x, x$$

```
> diff(1/x,x$5);
```

$$-\frac{120}{x^6}$$

Prázdná posloupnost:

```
> prazdna:=NULL;
```

$$prazdna :=$$

```
> whattype(prazdna);
```

$$exprseq$$

```
>
```

Datový typ seznam (list)

Záleží na pořadí prvků, prvky se v seznamu mohou vyskytovat vícekrát.

```
> p:=[seq(i^2,i=1..10)];
```

$$p := [1, 4, 9, 16, 25, 36, 49, 64, 81, 100]$$

```
> whattype(p);
```

$$list$$

```
> p[4];
```

$$16$$

```
> p[1..4];
```

$$[1, 4, 9, 16]$$

```
> nops(p);
```

$$10$$

Převod na typ posloupnost:

```
> op(p);
```

$$1, 4, 9, 16, 25, 36, 49, 64, 81, 100$$

```
> whattype(%);
```

$$exprseq$$

Přidání prvku do seznamu:

```
> p;
```

$$[1, 4, 9, 16, 25, 36, 49, 64, 81, 100]$$

```
> [op(p),dalsi_prvek];
```

$$[1, 4, 9, 16, 25, 36, 49, 64, 81, 100, dalsi\_prvek]$$

```
>
```

Datový typ množina (set)

Nezáleží na pořadí prvků.

```
> M:={1,2,3,1};
```

$$M := \{1, 2, 3\}$$

```
> whattype(M);
```

$$set$$

```
> S:=convert(M,list);
```

$$S := [1, 2, 3]$$

```
> whattype(S);
```

$$list$$

```
> convert(S,set);
```

$$\{1, 2, 3\}$$

```
> whattype(%);
```

$$set$$

```
> M:={seq(i^2,i=1..10)};
```

$$M := \{1, 4, 9, 16, 25, 36, 49, 64, 81, 100\}$$

```
> whattype(M);
```

$$set$$

Prázdná množina:

```
> M:={};
```

$$M := \{\ \}$$

```
> whattype(M);
```

set

```
>
```

Operace s množinami:

```
> A:={seq(2*i-1,i=6..50)};
```

$A := \{11, 13, 15, 17, 19, 21, 23, 25, 27, 29, 31, 33, 35, 37, 39, 41, 43, 45, 47, 49, 51, 53, 55, 57, 59, 61, 63, 65, 67, 69, 71, 73, 75, 77, 79, 81, 83, 85, 87, 89, 91, 93, 95, 97, 99\}$

(lichá dvojciferná čísla)

Jiná možnost:

```
> A:={seq(i,i=11..99,2)};
```

$A := \{11, 13, 15, 17, 19, 21, 23, 25, 27, 29, 31, 33, 35, 37, 39, 41, 43, 45, 47, 49, 51, 53, 55, 57, 59, 61, 63, 65, 67, 69, 71, 73, 75, 77, 79, 81, 83, 85, 87, 89, 91, 93, 95, 97, 99\}$

```
> evalb(10 in A);
```

false

```
> B:={seq(3*i+1,i=3..32)};
```

$B := \{10, 13, 16, 19, 22, 25, 28, 31, 34, 37, 40, 43, 46, 49, 52, 55, 58, 61, 64, 67, 70, 73, 76, 79, 82, 85, 88, 91, 94, 97\}$

(dvojciferná čísla dávající po dělení číslem 3 zbytek 1)

Sjednocení:

```
> A union B;
```

$\{10, 11, 13, 15, 16, 17, 19, 21, 22, 23, 25, 27, 28, 29, 31, 33, 34, 35, 37, 39, 40, 41, 43, 45, 46, 47, 49, 51, 52, 53, 55, 57, 58, 59, 61, 63, 64, 65, 67, 69, 70, 71, 73, 75, 76, 77, 79, 81, 82, 83, 85, 87, 88, 89, 91, 93, 94, 95, 97, 99\}$

Průnik:

```
> A intersect B;
```

$\{13, 19, 25, 31, 37, 43, 49, 55, 61, 67, 73, 79, 85, 91, 97\}$

Rozdíl:

```
> A minus B;
```

$\{11, 15, 17, 21, 23, 27, 29, 33, 35, 39, 41, 45, 47, 51, 53, 57, 59, 63, 65, 69, 71, 75, 77, 81, 83, 87, 89, 93, 95, 99\}$

Přidání prvku do množiny:

```
> X:=A union {x};
```

$X := \{11, 13, 15, 17, 19, 21, 23, 25, 27, 29, 31, 33, 35, 37, 39, 41, 43, 45, 47, 49, 51, 53, 55, 57, 59, 61, 63, 65, 67, 69, 71, 73, 75, 77, 79, 81, 83, 85, 87, 89, 91, 93, 95, 97, 99, x\}$

Odebrání prvku z množiny:

```
> X minus {x};
```

$\{11, 13, 15, 17, 19, 21, 23, 25, 27, 29, 31, 33, 35, 37, 39, 41, 43, 45, 47, 49, 51, 53, 55, 57, 59, 61, 63, 65, 67, 69, 71, 73, 75, 77, 79, 81, 83, 85, 87, 89, 91, 93, 95, 97, 99\}$

Výběr prvků podle určiých kritérií:

```
> B;
```

$\{10, 13, 16, 19, 22, 25, 28, 31, 34, 37, 40, 43, 46, 49, 52, 55, 58, 61, 64, 67, 70, 73, 76, 79, 82, 85, 88, 91, 94, 97\}$

```
> select(isprime,B);
```

$$\{13, 19, 31, 37, 43, 61, 67, 73, 79, 97\}$$

(vytvoří podmnožinu množiny B tvořenou pouze prvočísly)

```
> select(x->irem(x,5)=0, B);
```

$$\{10, 25, 40, 55, 70, 85\}$$

(vybere ty prvky z množiny B, které jsou dělitelné číslem 5)

```
> select(x->x<50,B);
```

$$\{10, 13, 16, 19, 22, 25, 28, 31, 34, 37, 40, 43, 46, 49\}$$

(vybere prvky menší než 50)

```
>
```

```
>
```

Typ float a chyby při výpočtech

Zamysleme se nad následujícími výpočty:

```
> 10^30 + 1 - 10^30;
```

$$1$$

Nyní místo celého čísla 1 (typ integer) použijeme číslo 1.0 (typ float - číslo s pohyblivou desetinnou čárkou)

```
> 10^30 + 1.0 - 10^30;
```

$$0.$$

Výsledek je pochopitelně špatně !!!

Je to způsobeno následujícím:

```
> 10^30 + 1.0;
```

$$0.1000000000\ 10^{31}$$

Čísla typu float jsou v Maplu ukládána ve tvaru: 0, *abcdefghij* x 10^u. Defaultně se mantisa bere s deseti platnými ciframi.

Náprava:

```
> Digits:=50;   ### počet platných cifer, na které Maple počítá
```

$$Digits := 50$$

```
> 10^30 + 1.0 - 10^30;
```

$$1.0$$

```
> 10^30 + 1.0;
```

$$0.1000000000000000000000000000010\ 10^{31}$$

```
>
```

Další příklady

Integrál

Budeme počítat integrál $\int_0^1 \mathbf{e}^x x^{23}\, dx$.

```
> restart;
> i:=int(exp(x)*x^23,x=0..1);
```

$$i := 25852016738884976640000 - 9510425471055777937262\ \mathbf{e}$$

Nyní vypočteme numerickou hodnotu integrálu.

```
> evalf(i);
```

$$0.1\ 10^{14}$$

```
> plot(exp(x)*x^23,x=0..1,filled=true);
```



Výsledek je evidentně chybný (obsah plochy zřejmě není $0.1\ 10^{14}$). Nyní se pokusíme vysvětlit, proč se tak stalo.

```
> i;
```

$$25852016738884976640000 - 95104254710555777937262\ \mathbf{e}$$

Ve výše uvedeném výrazu se odečítají dvě skoro stejná čísla.
V momentě, kdy zavoláme na výraz **i** příkaz **evalf**, se vše zkonvertuje na typ "float". Tím dojde k zaokrouhlovacím chybám, které mohou být značné.

```
> convert(25852016738884976640000,float);
```

$$0.2585201674\ 10^{23}$$

```
> evalf(95104254710555777937262*exp(1));
```

$$0.2585201673\ 10^{23}$$

```
> %%-%;
```

$$0.1\ 10^{14}$$

Řešení je snadné - stačí proměnnou "Digits" nastavit na větší hodnotu.

```
> Digits:=50;
```

$$\mathit{Digits} := 50$$

```
> i;
```

$$25852016738884976640000 - 95104254710555777937262\ \mathbf{e}$$

```
> evalf(i);
```

$$0.10889929110961911960343415 8$$

Můžeme už tomuto výsledku věřit?

```
>
```

Posloupnost

Je dána posloupnost a_n splňující následující rekurentní vztah:

$$a_0 = 1,\ a_1 = \frac{1}{11},\ a_{n+2} = \frac{34\,a_{n+1}}{11} - \frac{3\,a_n}{11}.$$

Chceme spočítat 50. člen této posloupnosti. Lze ukázat, že $a_n = \dfrac{1}{11^n}$ (například indukcí), a tedy 50. člen je roven $\dfrac{1}{11^{50}}$.

Přibližná hodnota a_{50}:

```
> evalf(1/11^50);
```

$$0.8518551280\ 10^{-52}$$

Nyní vytvoříme proceduru, která bude počítat členy naší posloupnosti pomocí rekurentního vztahu. Přitom budeme počítat numericky s přesností na d platných cifer.

```
> posl:=proc(n,d) local a,i;
> Digits:=d;
> a[0]:=1.0;
> a[1]:=1/11;
> for i from 2 to n do
> a[i]:=34/11*a[i-1]-3/11*a[i-2];
> od;
> end;
```

posl := **proc**(*n*, *d*)
local *a*, *i*;
Digits := *d*;
a[0] := 1.0;
a[1] := 1 / 11;
for *i* **from** 2 **to** *n* **do** *a*[*i*] := 34 / 11**a*[*i* − 1] − 3 / 11**a*[*i* − 2] **end do**
end proc

Prvních 50 členů spočítaných na 10 platných cifer:

```
> seq(posl(i,10),i=1..50);
```

0.0082644628, 0.00075131477, 0.000068301253, $0.62089356\ 10^{-5}$, $0.56364104\ 10^{-6}$, $0.48817142\ 10^{-7}$, $-0.28309356\ 10^{-8}$, $-0.2206393058\ 10^{-7}$, $-0.6742553026\ 10^{-7}$, $-0.2023887488\ 10^{-6}$, $-0.6071764426\ 10^{-6}$, $-0.1821530255\ 10^{-5}$, $-0.5464590850\ 10^{-5}$, -0.00001639377256, -0.00004918131768, -0.0001475439530, -0.0004426318590, -0.001327895577, -0.003983686731, -0.01195106019, -0.03585318057, -0.1075595417, -0.3226786251, -0.9680358753, -2.904107626, -8.712322878,

-26.13696864, -78.41090592, -235.2327177, -705.6981531, -2117.094459,
-6351.283377, -19053.85013, -57161.55039, -171484.6512, -514453.9536,
$-0.1543361861\ 10^{7}$, $-0.4630085583\ 10^{7}$, $-0.1389025675\ 10^{8}$, $-0.4167077025\ 10^{8}$,
$-0.1250123108\ 10^{9}$, $-0.3750369324\ 10^{9}$, $-0.1125110797\ 10^{10}$, $-0.3375332391\ 10^{10}$,
$-0.1012599717\ 10^{11}$, $-0.3037799151\ 10^{11}$, $-0.9113397453\ 10^{11}$, $-0.2734019236\ 10^{12}$,
$-0.8202057708\ 10^{12}$

```
>
```

50. člen spočítaný s přesností na 10, 20, 30, ... , 70, 80 platných cifer:

```
> seq(posl(50,10*i),i=1..8);
```

$-0.8202057708\ 10^{12}$, -162.11626006473896482,
$0.61320722376294679933082227559 4\ 10^{-7}$,
$0.511000536144293587835094332418427857848 0\ 10^{-17}$,
$0.43886001308718349469990854545787544404241258099099\ 10^{-27}$,
$0.369249279057482329886842015140475762971511860831137987022451\ 10^{-37}$, 0.3
0206272486493515244315492038023739513445473178200377053881964652721 85
10^{-47}, 0.8518572242020694571770124691506155530122239162023482056646045 53
$651864283169763\ 10^{-52}$

```
>
```

Banka

Představme si banku, do které vložíme e Kč (Eulerovo číslo). Po každém roce se z účtu odečte 1 Kč (vedení účtu) a zůstatek se vynásobí počtem let od založení účtu. Zjistěte, jak výhodná je nabídka, kterou nám banka dává. Kolik Kč budeme mít na účtu za 22 let ?

Je jasné, že se jedná o rekurzi $a_n = n\,(a_{n-1} - 1)$, kde $a_0 = \mathbf{e}$. a_n pak udává stav účtu po n letech od založení účtu.

Implementace:

```
> restart;

> banka:=proc(n) local i,a;
> a:=exp(1);
> for i from 1 to n do
> a:=i*(a-1);
> od;
> end;
```

$banka := \mathbf{proc}(n)\ \mathbf{local}\ i, a;\ a := \exp(1);\ \mathbf{for}\ i\ \mathbf{to}\ n\ \mathbf{do}\ a := i*(a-1)\ \mathbf{end\ do}\ \mathbf{end\ proc}$

```
> evalf(banka(22));
```

$$0.1\ 10^{13}$$

Za 22 let dostaneme tedy 1 bilión Kč. :-))) Okamžitě běžíme do banky uzavřít smlouvu ....

Výpočet na více platných cifer:

```
> evalf(banka(22),15);
```

$$0.1\ 10^{8}$$

```
> evalf(banka(22),20);
```

$$0.$$

Rozeberme (pro jistotu) celou situaci pouze s tužkou a papírem (a se znalostmi matematické analýzy).

Zkusíme ještě větší počet platných cifer:

```
> evalf(banka(22),50);
```

$$1.0453652129514255963919495415$$

```
>
```

Chyby v grafice

2D grafika:

```
> restart;
> f:=x->exp(x)+ln(abs(4-x));
```

$$f := x \rightarrow \mathbf{e}^{x} + \ln(|4-x|)$$

```
> plot(f,0..5);
```



Vysvětlení:

```
> plot(f,3.99..4.01);
```




```
> f(3.999);
```

$$47.63582389$$

3D grafika:

```
> restart;
> f:=sin(2*Pi*x)*sin(2*Pi*y);
```

$$f := \sin(2\,\pi\,x)\,\sin(2\,\pi\,y)$$

```
> plot3d(f,x=0..1,y=0..1,axes=framed);
```




```
> plot3d(f,x=0..2,y=0..2,axes=framed);
```




```
> plot3d(f,x=0..25,y=0..25,axes=framed);
```



Náprava:

```
> plot3d(f,x=0..25,y=0..25,axes=framed,grid=[100,100]);
```



Funkce zadaná implicitně:

```
> restart;
> with(plots):
> implicitplot((x-y)^2=0,x=-1..1,y=-1..1);
```




```
> implicitplot((x-y)^2=0,x=-1..1,y=-1..1,grid=[200,200]);
```



Zamyslete se, jak má tato funkce (daná implicitně) vypadat.

>

Cvičení

1) Vygenerujte množinu A obsahující prvních 100 prvočísel a množinu B obsahující prvních 100 přirozených čísel, která po dělení číslem 3 dávají zbytek 2. Určete sjednocení a průnik těchto množin. Kolik prvků má sjednocení a kolik průnik? Dále určet rozdíl A\B a rozdíl B\A.

2) Z množiny A definované v úloze 1) vyberte všechna taková čísla x, pro něž je číslo $x+2$ dělitelné číslem 7.
(*Nápověda: Použijte příkaz "select"*.)

3) Na hřišti je jistý počet dětí (méně než 300). Pokud bychom je chtěli seřadit do dvojstupu, jedno dítě by zbylo. Pokud je seřadíme do trojstupu, opět jedno dítě zbyde. Jestliže je však

chceme seřadit do pětistupu, jedno dítě nám bude chybět. Seřadíme-li je do sedmistupu, jedno opět zbyde. Kolik dětí je na hřišti?

4) Definujme množinu M

```
> M:={12,32/11,"Ahoj",1.8163,1233/1000,{1,2,3,4,5},3,<<1,2>|<3,2>>};
```

$$M := \{3, 12, \frac{32}{11}, \frac{1233}{1000}, 1.8163, \text{"Ahoj"}, \begin{bmatrix} 1 & 3 \\ 2 & 2 \end{bmatrix}, \{1, 2, 3, 4, 5\}\}$$

Následně (pomocí příkazu **`select`**) vyberte z množiny

a) všechny celočíselné prvky

b) všechny zlomky

5) Pomocí příkazu **`rand`** (syntaxi vyhledejte v nápovědě) vygenerujte seznam 200 náhodných přirozených čísel z intervalu <0, 100>. Ze seznamu poté odstraňte všechny duplicity. Kolik čísel vám v seznamu po tomto odstranění zůstalo?

6) Projděte následující příkazy:

```
> f:=(x,y)->333.75*y^6+x^2*(11*x^2*y^2-y^6-121*y^4-2)+5.5*y^8+x/(2*y);
```

$$f := (x, y) \rightarrow 333.75\, y^6 + x^2\, (11\, x^2\, y^2 - y^6 - 121\, y^4 - 2) + 5.5\, y^8 + \frac{1}{2}\, \frac{x}{y}$$

```
> f(77617.0,33096.0);
```

$$0.5\ 10^{28}$$

```
> f(77617,33096);
```

$$1.172603940$$

Proč jsme dostali různé výsledky?
Kterému z výše uvedených výsledků můžeme věřit?
Jak Maple donutíme, aby výsledek napsal ve tvaru zlomku (a tedy opravdu přesně)?

```
>
```